8.962 PS 3 Solution to Problem 5

First, we define our fundamental coordinates:

```
In[1]:= coord = {r, θ, ϕ}

Out[1]= {r, θ, ϕ}
```

And give the values of x in terms of r,θ,ϕ:

```
In[2]:= x = {r Sin[θ] Cos[ϕ], r Sin[θ] Sin[ϕ], r Cos[θ]}

Out[2]= {r Cos[ϕ] Sin[θ], r Sin[θ] Sin[ϕ], r Cos[θ]}
```

Here is the inverse Jacobian:

```
In[3]:= Jinv = Table[D[x[[i]], coord[[j]]], {i, 1, 3}, {j, 1, 3}] // FullSimplify

Out[3]= {{Cos[ϕ] Sin[θ], r Cos[θ] Cos[ϕ], -r Sin[θ] Sin[ϕ]},
         {Sin[θ] Sin[ϕ], r Cos[θ] Sin[ϕ], r Cos[ϕ] Sin[θ]}, {Cos[θ], -r Sin[θ], 0}}
```

And the Jacobian:

```
In[4]:= J = Inverse[Jinv] // FullSimplify
```

$$\text{Out[4]= } \left\{\{\text{Cos}[\phi]\,\text{Sin}[\theta],\ \text{Sin}[\theta]\,\text{Sin}[\phi],\ \text{Cos}[\theta]\},\ \left\{\frac{\text{Cos}[\theta]\,\text{Cos}[\phi]}{r},\ \frac{\text{Cos}[\theta]\,\text{Sin}[\phi]}{r},\ -\frac{\text{Sin}[\theta]}{r}\right\},\ \left\{-\frac{\text{Csc}[\theta]\,\text{Sin}[\phi]}{r},\ \frac{\text{Cos}[\phi]\,\text{Csc}[\theta]}{r},\ 0\right\}\right\}$$

Here is $(\Gamma^{\alpha'})_{\beta' \gamma'}$ computed from the transformation rule for Γ:

```
In[5]:= MatrixForm[Γ = Table[Sum[- D[J[[αp, γ]], coord[[βp]]] Jinv[[γ, γp]], {γ, 1, 3}],
            {αp, 1, 3}, {βp, 1, 3}, {γp, 1, 3}] // FullSimplify]

    General::spell1 : Possible spelling error: new symbol name "βp" is similar to existing symbol "αp". More…

    General::spell :
     Possible spelling error: new symbol name "γp" is similar to existing symbols {αp, βp, γ}. More…
```

Out[5]//MatrixForm=

$$\begin{pmatrix} \begin{pmatrix} 0 \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ -r \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ -r\,\text{Sin}[\theta]^2 \end{pmatrix} \\ \begin{pmatrix} 0 \\ \frac{1}{r} \\ 0 \end{pmatrix} & \begin{pmatrix} \frac{1}{r} \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ -\text{Cos}[\theta]\,\text{Sin}[\theta] \end{pmatrix} \\ \begin{pmatrix} 0 \\ 0 \\ \frac{1}{r} \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ \text{Cot}[\theta] \end{pmatrix} & \begin{pmatrix} \frac{1}{r} \\ \text{Cot}[\theta] \\ 0 \end{pmatrix} \end{pmatrix}$$

We know that the metric in cartesian coordinates is diagonal:

```
In[6]:= g = {{1, 0, 0}, {0, 1, 0}, {0, 0, 1}}

Out[6]= {{1, 0, 0}, {0, 1, 0}, {0, 0, 1}}
```

And we can transform that to get the metric in spherical coordinates:

```
In[7]:= gHat = Transpose[Jinv].g.Jinv // FullSimplify
```

Out[7]= $\{\{1, 0, 0\}, \{0, r^2, 0\}, \{0, 0, r^2 \,\mathrm{Sin}[\theta]^2\}\}$

Invert that to obtain the inverse metric in spherical coordinates.

```
In[8]:= gHatInv = Inverse[gHat]
```

Out[8]= $\left\{\{1, 0, 0\}, \left\{0, \frac{1}{r^2}, 0\right\}, \left\{0, 0, \frac{\mathrm{Csc}[\theta]^2}{r^2}\right\}\right\}$

Here we check that $\nabla_{i'} g_{j' k'} = 0$:

```
In[9]:= MatrixForm[Table[D[gHat[[j, k]], coord[[i]]] -
          Sum[Γ[[l, i, j]] gHat[[l, k]], {l, 1, 3}] - Sum[Γ[[l, i, k]] gHat[[j, l]], {l, 1, 3}],
         {i, 1, 3}, {j, 1, 3}, {k, 1, 3}]] // FullSimplify
```

Out[9]//MatrixForm=

$$\begin{pmatrix} \begin{pmatrix}0\\0\\0\end{pmatrix} & \begin{pmatrix}0\\0\\0\end{pmatrix} & \begin{pmatrix}0\\0\\0\end{pmatrix} \\ \begin{pmatrix}0\\0\\0\end{pmatrix} & \begin{pmatrix}0\\0\\0\end{pmatrix} & \begin{pmatrix}0\\0\\0\end{pmatrix} \\ \begin{pmatrix}0\\0\\0\end{pmatrix} & \begin{pmatrix}0\\0\\0\end{pmatrix} & \begin{pmatrix}0\\0\\0\end{pmatrix} \end{pmatrix}$$

And here we check that $(R_{a' b'}{}^{c'})_{d'} = 0$ (i.e. that euclidean space is flat, even in spherical coordinates:

```
In[10]:= Table[D[Γ[[c, b, d]], coord[[a]]] - D[Γ[[c, a, d]], coord[[b]]] +
           Sum[Γ[[c, a, e]] Γ[[e, b, d]] - Γ[[c, b, e]] Γ[[e, a, d]], {e, 1, 3}],
          {a, 1, 3}, {b, 1, 3}, {c, 1, 3}, {d, 1, 3}] // FullSimplify

Out[10]= {{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
           {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}},
          {{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
           {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}, {{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
           {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}}
```

We expect that $\nabla_{i'} f = \partial_{i'} f$, which is clearly not the gradient according to normal spherical vector calculus. But, we often treat the gradient as a vector, so we should really raise the index i':

```
In[11]:= Table[Sum[gHatInv[[ipp, ip]] D[f[r, θ, ϕ], coord[[ipp]]], {ipp, 1, 3}], {ip, 1, 3}] //
          FullSimplify
```

Out[11]= $\left\{f^{(1,0,0)}[r, \theta, \phi], \frac{f^{(0,1,0)}[r, \theta, \phi]}{r^2}, \frac{\mathrm{Csc}[\theta]^2 f^{(0,0,1)}[r, \theta, \phi]}{r^2}\right\}$

This is still not quite the formula for gradient that we're used to---there is an extra power of r and r Sin[θ] in the denominators of the second and third terms. The reason this formula differs from the standard one is that the basis vectors ∂_θ, and ∂_ϕ are not normalized. The extra powers of lengths which appear here and do not appear in the standard formula for the gradient in spherical coordinates are taking care of the normalization of basis vectors. In the standard formula, you are assumed to already be using normalized basis vectors.

Here is $\nabla_\mu V^\mu$:

```
In[12]:= With[{V = {Vr[r, θ, ϕ], Vθ[r, θ, ϕ], Vϕ[r, θ, ϕ]}},
          Sum[D[V[[μ]], coord[[μ]]] + Sum[Γ[[μ, μ, ν]] V[[ν]], {ν, 1, 3}], {μ, 1, 3}] //
           FullSimplify]

        General::spell1 : Possible spelling error: new symbol name "Vϕ" is similar to existing symbol "Vθ". More…
```

$$\text{Out[12]=}\ \frac{2\,\text{Vr}[r,\theta,\phi]}{r} + \text{Cot}[\theta]\,\text{V}\theta[r,\theta,\phi] + \text{V}\phi^{(0,0,1)}[r,\theta,\phi] + \text{V}\theta^{(0,1,0)}[r,\theta,\phi] + \text{Vr}^{(1,0,0)}[r,\theta,\phi]$$

Again, this doesn't look like the standard formulas because the basis vectors are not normalized.

Here is the Levi-Civita tensor for computing ∇×V:

```
In[22]:= ϵ = Table[Sqrt[Det[gHat]] Signature[{i, j, k}], {i, 1, 3}, {j, 1, 3}, {k, 1, 3}] //
           FullSimplify[#, {r > 0, Sin[θ] > 0}] &
```

$$\text{Out[22]=}\ \{\{\{0,0,0\},\{0,0,r^2\,\text{Sin}[\theta]\},\{0,-r^2\,\text{Sin}[\theta],0\}\}, \{\{0,0,-r^2\,\text{Sin}[\theta]\},\{0,0,0\},\{r^2\,\text{Sin}[\theta],0,0\}\}, \{\{0,r^2\,\text{Sin}[\theta],0\},\{-r^2\,\text{Sin}[\theta],0,0\},\{0,0,0\}\}\}$$

And here is $(\epsilon_i)^{jk}\,\nabla_j V_k$:

```
In[28]:= With[{V = {Vr[r, θ, ϕ], Vθ[r, θ, ϕ], Vϕ[r, θ, ϕ]},
           ϵ = Evaluate[Table[Sum[gHatInv[[k, kp]] gHatInv[[j, jp]] ϵ[[i, j, k]],
               {j, 1, 3}, {k, 1, 3}], {i, 1, 3}, {jp, 1, 3}, {kp, 1, 3}]]}, FullSimplify[
           Table[Sum[ϵ[[i, j, k]] (D[V[[k]], coord[[j]]] - Sum[Γ[[l, j, k]] V[[l]], {l, 1, 3}]),
             {j, 1, 3}, {k, 1, 3}], {i, 1, 3}]]]
```

$$\text{Out[28]=}\ \left\{\frac{\text{Csc}[\theta]\,(-\text{V}\theta^{(0,0,1)}[r,\theta,\phi] + \text{V}\phi^{(0,1,0)}[r,\theta,\phi])}{r^2},\ \text{Csc}[\theta]\,(\text{Vr}^{(0,0,1)}[r,\theta,\phi] - \text{V}\phi^{(1,0,0)}[r,\theta,\phi]),\ \text{Sin}[\theta]\,(-\text{Vr}^{(0,1,0)}[r,\theta,\phi] + \text{V}\theta^{(1,0,0)}[r,\theta,\phi])\right\}$$

Again, it's different because our coordinate basis is not orthonormal (and we still have a lower i index).